



# HOKKAIDO UNIVERSITY

Title	時系列データ解析のためのBASICプログラム
Author(s)	石田, 完; ISHIDA, Tamotsu
Citation	低温科学. 物理篇. 資料集, 40, 87-113
Issue Date	1982-03-05
Doc URL	<a href="https://hdl.handle.net/2115/18723">https://hdl.handle.net/2115/18723</a>
Type	departmental bulletin paper
File Information	40_p87-113.pdf



## 時系列データ解析のためのBASICプログラム\*

石田 完

(低温科学研究所)

(昭和56年12月受理)

### I. まえがき

最近各研究室毎にマイクロ・コンピュータを備え、データの集取、解析に利用するようになってきた。マイクロ・コンピュータの使い方としては、実験装置を制御し、実時間でデータ集取を行なうのに適しているが、インテリジェント端末として、通信回線により大型計算機に接続し、計算は大型機で行なわれて、データの前処理、結果の後処理に利用することも行なわれている。

しかしメモリの増大に伴って、かなり高度な数値計算をBASICプログラムで実行できるようになった。IF 800を導入以来、多数のプログラムが蓄積されたので、ここでは汎用性があると思われる主に時系列データの解析プログラムを紹介する。プログラムはすべてIF 800用のOKI-BASICで書いてあるが、僅かの変更でどの機種にも使える筈である。

### II. FFT法によるスペクトルと相関関数

Program 1～4はすべてコンプリート・プログラムで、Program 5はProgram 1, 3で用いるFFTサブルーチン・プログラム、Program 6はProgram 2で用いるWINDOWサブルーチン・プログラムである。必要な入力パラメータはすべて会話型で問うてくるので、キーボードから入力すればよい。Program 1から4まで順次実行させれば2組の時系列データのパワースペクトル、クロススペクトル、自己相関関数、相互相関関数、コヒーレンス、位相角が順次求められる。メモリ容量の制限から、手順を4個の独立プログラムに分けたが、データ数が少なければ、これらを連結して一つのプログラムとすることもできる。

**Program 1:** 2組のデータのFFTによるフーリエ変換。2組のデータは夫々Xnnn.DAT, Ynnn.DATと云うファイル名でドライブ2のフロッピーディスクにはいっていないなければならない。nnnは1～3桁の整数で、データ番号としてプログラムがきいてくる。データ数の上限は512としてある。FFTを用いるので、その数は2のべき乗でなければならず、256, 128, 64等である。結果の実数部、虚数部はX, Yデータに対して、AXnnn.DAT, BXnnn.DAT, AYnnn.DAT, BYnnn.DAT

\*北海道大学低温科学研究所業績 第2413号

なるファイル名でドライブ2のフロッピーディスクに格納される。その個数は初めの入力データの数を $N$ とすると、夫々 $N/2+1$ 個である。

**Program 2** : Program 1の結果からハミング・ウィンドウで平滑化したパワースペクトルとクロススペクトルを計算する。入力パラメータは入力原データ $X, Y$ の個数 $N$ , データ番号 $nnn$ , データのサンプリング間隔 $\Delta t$  sec, ハミング・ウィンドウの繰りかえし数である。結果のパワースペクトルは $PXnnn.DAT, PYnnn.DAT$ のファイル名で、クロススペクトルのコスベクトル, クオドスペクトルは $KOnnn.DAT, QUnnn.DAT$ のファイル名でドライブ2のフロッピーディスクに格納される。夫々のデータ数は $N/2+1$ である。

**Program 3** : Program 2の結果のコスベクトル, クオドスペクトルから原データの自己相関関数と相互相関関数を計算する。入力パラメータは原データ数の $1/2$ , すなわち $N/2$ , データ番号 $nnn$ , 原データのサンプリング間隔 $\Delta t$  secである。自己相関関数は $CXnnn.DAT, CYnnn.DAT$ のファイル名で夫々 $N/4+1$ 個, 相互相関関数は $CXYnnn.DAT$ のファイル名で $N/2+1$ 個, ドライブ2のフロッピーディスクに格納される。相互相関関数データの前半はラグ負, 後半はラグ正に相当し,  $N/4+1$ 番目のデータがラグ0に相当する。ラグ間隔は計算終了時に $DTAU=……$ と秒単位で表示される。

**Program 4** : Program 2の結果からコヒーレンスと位相角が計算される。入力パラメータは原データの個数 $N$ , データ番号 $nnn$ , サンプリング間隔 $\Delta t$  secである。コヒーレンスと位相角は $COHnnn.DAT, PHAnnn.DAT$ のファイル名で夫々 $N/2+1$ 個, ドライブ2のフロッピーディスクに格納される。位相角は $-\pi$ から $+\pi$ の範囲になるように調整され,  $2\pi$ で規格化される。

Program 5のFFTは後の例に示すように各種のプログラムに組込むことができる。Program 6のウィンドウ・サブルーチン・プログラムではパラメータ $KKK=1$ でHanning window,  $KKK=2$ でHamming windowとなる。以上のプログラムの算法は文献1)に従った。

### III. 最大エントロピー法 (MEM) によるスペクトル解析

**Program 7**がMEMのサブルーチン・プログラムであり, その使用例を**Program 8**に示す。

パラメータ: Program 8の行番号40, 50に示される。

$N$ : 入力データの個数

$LAG$ : 自己相関関数の最適ラグ数。  $LAG \approx (2 \sim 3) \sqrt{N}$ で見積られるが予測誤差の期待値 (final prediction error)  $FPE$ が計算されるので, その最小値の項数で打切るべきである。

$LMAX$ : 最大ラグ数。自己相関関数は $LMAX$ まで計算されるが,  $LAG$ 以上の部分は誤差が大きい。

$FMIN$ : 求めるスペクトルの最小周波数 (Hz)

$FMAX$ : 求めるスペクトルの最大周波数 (Hz)

$DF$ : スペクトルの分解巾 (Hz)

DT：データのサンプリング間隔 (sec)

入力データは配列X(N)として与え、結果のパワースペクトルは周波数配列F(K)に対して配列S(K)に求められ自己相関関数はDT間隔でC(LMAX)に求まる。なお行番号20, 30に示すように作業用の配列を与えたパラメータの大きさに従ってA(LAG), B1(N), B2(N), G(LAG), FPE(LAG), F(K)を宣言しなければならない。

スペクトルの最小最大周波数の限界はもち論0および $1/(2 * DT)$  Hzであるが、その範囲内であれば任意に与えることができる。それは特定周波数成分を求めるフーリエ変換(行番号4000-4050のサブルーチン)を用いているためであるが、このルーチンは漸化式を用いているので、精度を上げるため倍精度演算を行なっている。周波数間隔(スペクトル分解巾)DFを任意に与えることにより求められるスペクトルの項数は

$$K = \text{INT} ((FMAX - FMIN) / DF + 1.5) \quad (1)$$

と計算される。

Program 8では行番号70-100に示すように、入力データとして周波数0.15 Hzの余弦波に振巾が2倍で周波数0.2 Hzの余弦波を加え、更に振巾0.6倍のランダムノイズを加えた波形を与えたが、僅か25個のデータで、結果は明らかに二つのピークを0.15 Hzと0.2 Hzとに分離している。このようにMEMは少い個数のデータから、狭い分解巾でスペクトルを求めうる特徴があるが、スペクトルのピーク値が直接パワーを与えない欠点もある。Program 7の算法は文献1)に従った。

#### IV. デジタル・フィルタ

時系列波形を処理する場合、まづアナログフィルタを通してAD変換をするが、その後も任意の周波数特性を持つデジタルフィルタを通すことはデータの前処理として重要である。

**Program 9**：次の如きフィルタのZ変換の係数A(i, j), B(i, j)を求めるサブルーチン・プログラム。

$$F(z) = V \cdot V \cdot \prod_{j=1}^k \frac{A(1, j) + A(2, i)Z + A(3, j)Z^2}{B(1, j) + B(2, j)Z + B(3, j)Z^2}$$

パラメータ：

PAS\$：ローパスの時L, ハイパスの時Hとする。

CF：カットオフ周波数 (Hz)

HF：パワーが $\frac{1}{2}$ に落ちる周波数 (Hz)

DT：フィルタしようとする時系列データのサンプリング間隔 (sec)

D0：パスバンドに於けるリプル巾, パワー比で与える時は正の実数, デンベルで与える時は負の実数とする。

以上が与えるパラメータで以下はこのルーチンで計算される。

KK%：計算されるフィルター係数配列A(i, j), B(i, j)のjの上限值。

VV：(2)式に於けるフィルタの増巾係数。

A(), B()：計算されるフィルタの係数配列でDIM A(3, 6), B(3, 6)と宣言するこ

とが必要。

このプログラムの算法は文献2)に従った。

**Program 10** : Program 9 で求められるフィルタの周波数特性を計算するサブルーチン・プログラム。

パラメータ :

A ( ), B ( ) : Program 9 で計算された係数の配列。

KK% : Program 9 で計算された係数配列の2番目添字の上限値。

RESP ( ) : 計算されるレスポンスの配列。

PHASE ( ) : 計算される位相の配列。 $-\pi$  から  $+\pi$  の範囲にはいるように調整され、 $2\pi$  で規格化される。

DT : フィルタしようとする時系列データのサンプリング間隔 (sec)。

VV : Program 9 で計算されるフィルタの増巾係数。

N : レスポンス, 位相配列の添字の上限値で, 求めたい周波数間隔を DF (Hz) とすれば  $N = \text{INT} (\frac{1}{2} * DF * DT) + 1.5$  である。

**Program 11** : Program 9 の(2)式を用いて

$$Y (Z) = F (Z) \cdot X (Z) \quad (3)$$

により, 入力Xから出力Yを計算するサブルーチン・プログラム。

パラメータ :

X ( ) : 入力データの配列。計算終了時にはフィルタされた結果がはいる。

N : 入力データの個数。

A ( ), B ( ) : Program 9 で求めたフィルタ係数の配列。

KK% : A, B配列の2番目添字の上限値 (Program 9 で計算される)。

VV : フィルタの増巾係数 (Program 9 で計算される)。

DIR\$ : Pのとき正方向にフィルタ ( $X_1, X_2, \dots, X_N \rightarrow Y_1, Y_2, \dots, Y_N$ )。Nのとき逆方向にフィルタ ( $X_N, X_{N-1}, \dots, X_1 \rightarrow Y_1, Y_2, \dots, Y_N$ )。

初めDIR\$ = "P" として計算し, その結果をDIR\$ = "N" として再び計算すれば, その結果は全く位相変化0のフィルタ出力となる。Program 9 でハイパス・フィルタとローパス・フィルタの係数を求めておき, Program 11 を2度用いれば, バンドパスとか特定周波数のリジェクトを行なうことができる。

**Program 12** : Program 9, 10 の使用例。CF=60 Hz, HF=62 Hz, D0=0.5 dB のローパス・フィルタを設計し, その周波数特性を1 Hz 間隔で求めた。結果の第1図は後に示す Program 15 で描かせた。

## V. 図形表示プログラム

計算処理された結果はグラフ化するのが通常である。Program 13 を除いて, 以下のプログラムは

図形をCRTに出力し、そのハードコピーをとるものである。IF 800はかなり分解能が良いので、そのハードコピーは充分原稿図になり得る。ただしIF 800のディスプレイでは表示文字数が1行80文字、25行であり、ドット表示は横640、縦200ドットとなっているので、数値情報をグラフ化する場合、縦横のスケーリング、また文字を適当な位置に表示させるのに特別の配慮が必要である。従ってプリント型式のProgram 13以外は他機種に適用する場合はかなりの変更を要する。

**Program 13:** 数値データ群の頻度分布図をプリントするコンプリート・プログラム。入力情報としてはデータのはいつているファイル名、データの個数、データの名前、データのユニット名をキーインする。頻度のクラス巾Wは、データの最大値AMAX、最小値AMINが求められ、データ個数をNとすると

$$W = (AMAX - AMIN) / (N/20) \quad (4)$$

と定められ、小数点以下2桁目で4捨5入される。そのため入力データはすべて、最大値が小数点以上2桁の数値となるように行番号600以下のサブルーチンで調整される。10の何乗されたかはUnit=……と表示される。分布図の表示は50%までで、それをこすとOverと表示される(数値は正しく表示される)。

第2図にこの使用例を示す。

**Program 14:** 1組のデータのグラフ表示。入力情報は入力データのはいつているファイル名、データ個数(最大512)とデータ間隔、その間隔の単位名である。データが入力されると、表示するデータの範囲を問うてくる。データの番号で答える。第1番目は0である。データ個数が256とし、全範囲を表示させるなら0,255とキーインする。図が表示された後、Continue or Copy (Y/N/C)?と問うてくるので、コピーするならC、表示しなおすならY、終了するならNをキーインする。Yをキーインしていれば、データの任意の範囲を任意に拡大表示することができる。第3図がこの使用例で、MEMの使用例Program 8の結果を表示させた。

**Program 15:** 2組のデータのグラフ表示。2段にグラフ表示されることを除いて、Program 14と全く同じである。この使用例は第1図である。

**Program 16:** 両対数のグラフ表示。パワースペクトルはパワーも周波数も対数軸で表示することが多いので、そのために作った。両軸とも目盛は $(1, 2, 5) \times 10^p$ で示される。行番号3000以下のサブルーチンは任意の数値Xを小数点以上1桁の数に表現するときのPを求める計算である。この使用例を第4図に示す。データは第3図と共にProgram 8のMEMのテスト結果である。

**Program 17:** 等高線を描くコンプリート・プログラム<sup>3)</sup>。正方格子点上に $N \times M$ 個のデータが与えられたとき、指定した最大値AMAX、最小値AMINの間に、指定したND本の等高線を描く。データの真の最大、最小値を $A_m$ 、 $A_n$ 、等高線の数値間隔をDAとした時、

$$DA = \frac{AMAX - ANIN}{ND}, \quad AMAX < A_m + DA, \quad AMIN \leq A_n \quad (5)$$

の関係が満たされるように、AMAX、AMIN、NDを与えなければならない。第5図が使用例で

データは図の上部に示してある。真の最大，最小値は 6.91, -1.25 であるが， $AMAX=6$ ， $AMIN=-1$ ， $ND=10$  を与えた。データ配列  $A(I, J)$  の順序は図に示されるように， $I$  が横の列， $J$  が縦の列で図の原点は左下隅となる。

## VI. FFT を使用する時の注意

Program 18, 19 にサブルーチン・プログラム FFT と FTRAN の使用例を示した。FFT ではデータのサンプリング間隔  $\Delta t$  とデータ数  $N$  がきまれば，観測時間は  $T = \Delta t \cdot N$  とみなされ，求められる周波数間隔は  $\Delta f = 1/T$  となる。Program 18 の場合， $\Delta t = 1 \text{ sec}$  とすると  $N = 16$  で  $T = 16 \text{ sec}$ ， $\Delta f = 0.0625 \text{ Hz}$  である。従って結果を合せるため Program 19 の場合， $\Delta f = 0.0625 \text{ Hz}$  とした。なお FTRAN の場合，与える周波数間隔 DFF は無次元周波数で  $DFF = \Delta f \cdot \Delta t$  である。

FFT で求められるデータ数  $N$  の係数  $A(1) \sim A(N/2)$ ， $B(1) \sim B(N/2)$  はフーリエ変換の係数  $a_i$ ， $b_i$  と次の対応をしていることに注意しなければならない。

$$\begin{aligned} a_0 &= \frac{1}{T} A(1), \quad a_1 = \frac{2}{T} A(2) \cdots a_{N/2} = \frac{2}{T} B(1) \\ b_1 &= \frac{2}{T} B(2) \cdots b_{N/2-1} = \frac{2}{T} B(N/2) \end{aligned}$$

FTRAN の場合は  $I = 1, 2, \dots$  に従って順次  $CC, SS$  を求めてゆくので， $A(1) = CC_{I=1}$ ， $B(1) = SS_{I=1}$  とすれば

$$\begin{aligned} a_0 &= \frac{1}{T} A(1), \quad a_1 = \frac{2}{T} A(2), \dots a_{N/2} = \frac{2}{T} A\left(\frac{N}{2} + 1\right) \\ B(1) &= 0, \quad b_1 = \frac{2}{T} B(2), \dots b_{N/2-1} = \frac{2}{T} B\left(\frac{N}{2}\right), \quad B\left(\frac{N}{2} + 1\right) = 0 \end{aligned}$$

である。

FTRAN は高次の SIN, COS を求めるのに漸化式によっているので，精度を上げるため，倍精度演算を用いている。計算速度は FFT より数倍おそいが，任意の周波数に対するフーリエ係数を求める場合に便利である。求める周波数の最小，最大を FMIN, FMAX とすれば得られる係数の項数は(1)式で与えられる。もちろん  $FMAX \leq 1/2 \cdot \Delta t$  でなければならない。

## 文 献

- 1) 日野幹雄：スペクトル解析（朝倉書店，1977）
- 2) 芦田謙・斉藤正徳 1970 デジタル・チェビシェフ・フィルタ，物理探鉱，23，6-19.
- 3) 石田完 1979 等高線を描くプログラム CONTAO，北海道大学大型計機センター・ニュース，11，2，29-40.

## Program 1

```

LIST
10 '*** FOURIE CONVERSION ***   "FOURIE.XY"
20 DIM X(512),A(256),B(256),S(128)
30 INPUT "Number of data (512,256,128,64)";N
35 INPUT "Run No. (0-999)";NO : M$=MID$(STR$(NO),2,3)
40 N2=N/2 : N4=N/4 : NP=LOG(N)/LOG(2)
50 FOR I=1 TO N4-1 : S(I)=SIN(I*6.28319/N) : NEXT
60 CHAIN MERGE "FFT.SUB",70,ALL
70 D$="2:X"+M$+".DAT"
80 GOSUB 200 :PRINT "Mean X:";XM
90 A$="2:AX"+M$+".DAT" : B$="2:BX"+M$+".DAT"
100 GOSUB 300 :PRINT "X end",TIME$
110 D$="2:Y"+M$+".DAT"
120 GOSUB 200 :PRINT "Mean Y:";XM
130 A$="2:AY"+M$+".DAT" : B$="2:BY"+M$+".DAT"
140 GOSUB 300 :PRINT "Y end",TIME$
150 END
200 '*** FFT ***
210 OPEN D$ FOR INPUT AS #1
220 FOR I=1 TO N : INPUT #1,X(I) : NEXT : CLOSE #1
230 SX=0 : FOR I=1 TO N : SX=SX+X(I) : NEXT
240 XM=SX/N : FOR I=1 TO N : X(I)=X(I)-XM : NEXT
250 GOSUB 1000 : '--- FFT.SUB
260 FOR I=0 TO N2-1 : A(I)=A(I+1) : NEXT : A(N2)=B(1)
270 FOR I=1 TO N2-1 : B(I)=B(I+1) : NEXT : B(0)=0:B(N2)=0
280 RETURN
300 '*** OUTPUT OF RESULTS ***
310 OPEN A$ FOR OUTPUT AS #1
320 FOR I=0 TO N2 : PRINT #1,A(I) : NEXT :CLOSE #1
330 OPEN B$ FOR OUTPUT AS #1
340 FOR I=0 TO N2 : PRINT #1,B(I) : NEXT :CLOSE #1
350 RETURN
ok

```

## Program 2

```

LIST
10 '*** Power SPedtrum *** "POWSPC.SDC"
20 DIM X(513),A(256),B(256),C(256),D(256)
30 DEF FNP(F1,F2,F3,F4)=(F1*F2+F3*F4)*TN
40 INPUT "Number of data (512,256,128,64)";N
45 INPUT "Run No. (0-999)";NO : M$=MID$(STR$(NO),2,3)
46 INPUT "Data interval (sec)";DT
47 INPUT "Repeat No. of Window";NH
50 N2=N/2 : KKK=2
55 A$="2:AX"+M$+".DAT" : B$="2:BX"+M$+".DAT"
56 C$="2:AY"+M$+".DAT" : D$="2:BY"+M$+".DAT"
57 E$="2:PX"+M$+".DAT" : F$="2:PY"+M$+".DAT"
58 G$="2:KO"+M$+".DAT" : H$="2:QU"+M$+".DAT"
60 TN=DT/N : PRINT "START",TIME$
70 CHAIN MERGE "WINDOW.SUB",80,ALL
80 OPEN A$ FOR INPUT AS #1
90 OPEN B$ FOR INPUT AS #2
100 OPEN C$ FOR INPUT AS #3
110 OPEN D$ FOR INPUT AS #4
120 FOR I=0 TO N2 : INPUT #1,A(I) : NEXT
130 FOR I=0 TO N2 : INPUT #2,B(I) : NEXT
140 FOR I=0 TO N2 : INPUT #3,C(I) : NEXT
150 FOR I=0 TO N2 : INPUT #4,D(I) : NEXT
160 CLOSE #1,#2,#3,#4
170 FOR I=0 TO N2 : X(I)=FNP(A(I),A(I),B(I),B(I)) : NEXT
180 GOSUB 2000 : '--- WINDOW.SUB
190 OPEN E$ FOR OUTPUT AS #1
200 FOR I=0 TO N2 : PRINT #1,X(I); :NEXT :CLOSE #1
205 PRINT "PX end",TIME$
210 FOR I=0 TO N2 : X(I)=FNP(C(I),C(I),D(I),D(I)) : NEXT
220 GOSUB 2000 : '--- WINDOW.SUB
230 OPEN F$ FOR OUTPUT AS #1
240 FOR I=0 TO N2 : PRINT #1,X(I); : NEXT : CLOSE #1
245 PRINT "PY end",TIME$
250 FOR I=0 TO N2 : X(I)=FNP(A(I),C(I),B(I),D(I)) : NEXT
260 GOSUB 2000 : '--- WINDOW.SUB
270 OPEN G$ FOR OUTPUT AS #1
280 FOR I=0 TO N2 : PRINT #1,X(I); : NEXT : CLOSE #1
285 PRINT "KO end",TIME$
290 FOR I=0 TO N2 : X(I)=FNP(B(I),C(I),A(I),-D(I)) : NEXT
300 GOSUB 2000 : '--- WINDOW.SUB
310 OPEN H$ FOR OUTPUT AS #1
320 FOR I=0 TO N2 : PRINT #1,X(I); : NEXT
325 PRINT "QU end",TIME$ : PRINT "DF=";1/(DT*N);"Hz"
330 END
Ok

```

## Program 3

```

LIST
10 '*** Auto- and Cross-Correlation *** "CORREL.SOC"
20 DIM X(256),A(128),B(128),CXV(128),CYX(128),S(64)
30 INPUT "Number of data (256,128,64,32)";N
35 INPUT "Run No. (0-999)";NO : M$=MID$(STR$(NO),2,3)
36 INPUT "Sampling interval (sec)";DT
40 N2=N/2 : N4=N/4 : NP=LOG(N)/LOG(2)
50 T=2*DT*N : DF=1/T : DF2=2*DF
55 PRINT "START",TIME$
60 FOR I=1 TO N4-1 : S(I)=SIN(I*6.28319/N) : NEXT
70 CHAIN MERGE "FFT.SUB",80,ALL
80 D$="2:PX"+M$+".DAT"
90 GOSUB 300
100 A$="2:CX"+M$+".DAT"
110 GOSUB 400
115 PRINT "CX end",TIME$
120 D$="2:PV"+M$+".DAT"
130 GOSUB 300
140 A$="2:CV"+M$+".DAT"
150 GOSUB 400
155 PRINT "CV end",TIME$
156 A$="2:K0"+M$+".DAT":B$="2:QU"+M$+".DAT":C$="2:CVX"+M$+".DAT"
160 OPEN A$ FOR INPUT AS #1
170 FOR I=1 TO N : INPUT #1,X(I) : NEXT : CLOSE #1
180 GOSUB 1000 : '--- FFT.SUB
190 FOR I=0 TO N2-1 : CXV(I)=A(I+1) : NEXT : CXV(N2)=B(1)
195 PRINT "CXV end",TIME$
200 OPEN B$ FOR INPUT AS #1
210 FOR I=1 TO N : INPUT #1,X(I) : NEXT : CLOSE #1
220 GOSUB 1000 : '--- FFT.SUB
230 FOR I=1 TO N2-1 : CVX(I)=B(I+1) : NEXT : CVX(0)=0:CVX(N2)=0
235 PRINT "CVX end",TIME$
240 OPEN C$ FOR OUTPUT AS #1
250 FOR I=N2 TO 0 STEP -1
260 PRINT #1,(CXV(I)+CVX(I))*DF2 : NEXT : CLOSE #1
270 OPEN C$ FOR APPEND AS #1
280 FOR I=1 TO N2 : PRINT #1,(CXV(I)-CVX(I))*DF2 : NEXT
290 PRINT "DTAU=";2*DT
295 PRINT "CROSS-CORR end",TIME$ : END
300 '*** FFT1 ***
310 OPEN D$ FOR INPUT AS #1
320 FOR I=1 TO N : INPUT #1,X(I) : NEXT : CLOSE #1
330 GOSUB 1000
340 FOR I=0 TO N2-1 : R(I)=A(I+1) : NEXT : R(N2)=B(1)
350 RETURN
400 '*** OUTPUT1 ***
410 OPEN A$ FOR OUTPUT AS #1
420 FOR I=0 TO N2 : PRINT #1,R(I)*DF2 : NEXT : CLOSE #1
430 RETURN
Ok

```

## Program 4

```

LIST
10 '*** Coherence and Phase *** "COHERE.SOC"
20 DIM PX(256),PY(256),KO(256),QU(256),W(256)
30 INPUT "Number of data (512,256,128,64)";N
35 INPUT "Run No. (0-999)";NO : M$=MID$(STR$(NO),2,3)
36 INPUT "Sampling interval (sec)";DT
40 N2=N/2
50 T=DT*N : DF=1/T : P0=.00001
55 PRINT "START",TIME$
56 A$="2:PX"+M$+".DAT" : B$="2:PY"+M$+".DAT"
57 C$="2:KO"+M$+".DAT" : D$="2:QU"+M$+".DAT"
58 E$="2:COH"+M$+".DAT" : F$="2:PHA"+M$+".DAT"
60 OPEN A$ FOR INPUT AS #1
70 OPEN B$ FOR INPUT AS #2
80 OPEN C$ FOR INPUT AS #3
90 OPEN D$ FOR INPUT AS #4
100 FOR I=0 TO N2 : INPUT #1,PX(I) : NEXT
110 FOR I=0 TO N2 : INPUT #2,PY(I) : NEXT
120 FOR I=0 TO N2 : INPUT #3,KO(I) : NEXT
130 FOR I=0 TO N2 : INPUT #4,QU(I) : NEXT
140 CLOSE #1,#2,#3,#4
150 FOR I=1 TO N2 : PXV=(KO(I)^2+QU(I)^2) : PP=PX(I)*PY(I)
155 IF PXV<P0 AND PP<P0 THEN W(I)=0 ELSE W(I)=PXV/PP
156 NEXT I
160 OPEN E$ FOR OUTPUT AS #1
165 FOR I=0 TO N2 : PRINT #1,W(I); : NEXT : CLOSE #1
166 PRINT "COH end",TIME$
170 X0=1E-30 : PI=3.14159 : PI2=PI*2 : P2=PI/2
180 FOR I=0 TO N2
190 IF ABS(KO(I))<X0 AND QU(I)>0 THEN W(I)=P2 : GOTO 240
200 IF ABS(KO(I))<X0 AND QU(I)<0 THEN W(I)=-P2 : GOTO 240
210 IF KO(I)<0 AND QU(I)>0 THEN W(I)=PI+ATN(QU(I)/KO(I)) : GOTO 240
220 IF KO(I)<0 AND QU(I)<0 THEN W(I)=ATN(QU(I)/KO(I))-PI : GOTO 240
230 W(I)=ATN(QU(I)/KO(I))
240 NEXT I
245 OPEN F$ FOR OUTPUT AS #1
250 FOR I=0 TO N2 : PRINT #1,W(I)/PI2; : NEXT
255 PRINT "PHA end",TIME$
256 PRINT "DF=";DF;"Hz"
260 END
Ok

```

## Program 5

```

LIST
1000 '*** SUBROUTINE FFT ***
1010 FOR I=1 TO N2:II=I+N2
1020 A(I)=X(I)+X(II):B(I)=X(I)-X(II):NEXT I
1030 FOR I=1 TO N2:II=2*I
1040 X(II-1)=A(I):X(II)=B(I):NEXT I
1050 IP=N4:M1=1:NP1=NP-1
1060 FOR M=1 TO NP1:M2=2*M1:M0=M1-1
1070 FOR I=1 TO IP:I1=M2*(I-1)+1:I2=I1+M1:I3=I1+N2:I4=I3+M1
1080 A(I1)=X(I1)+X(I3):B(I1)=X(I1)-X(I3):A(I2)=X(I2):B(I2)=X(I4)
1090 IF M=1 THEN 1140
1100 FOR K=1 TO M0:K1=IP*K:K2=N4-K1:K3=I3+K:K4=I4+K
1110 S1=S(K2)*X(K3)-S(K1)*X(K4):S2=S(K1)*X(K3)+S(K2)*X(K4)
1120 I5=I1+K:I6=I1-K+M2:I7=I2+K
1130 A(I5)=X(I5)+S1:A(I6)=X(I5)-S1:B(I5)=X(I7)+S2:B(I6)=-X(I7)+S2:NEXT K
1140 NEXT I
1150 FOR J=1 TO IP:J1=M2*(J-1):J2=2*J1:J3=J2+M2
1160 FOR K=1 TO M2:J1K=J1+K:J2K=J2+K:J3K=J3+K:X(J2K)=A(J1K):X(J3K)=B(J1K)
1170 NEXT K,J
1180 IF IP=1 THEN 1200
1190 M1=M2:IP=IP#2:NEXT M
1200 RETURN
Ok

```

## Program 6

```

LIST
2000 '*** WINDOW.SUB ***
2010 ' KKK=1:Hanning,KKK=2:Hamming,NH:Number of rePeat
2020 IF NH<1 THEN RETURN
2030 ON KKK GOTO 2040,2050
2040 CC1=.25 : CC2=.5 : GOTO 2060
2050 CC1=.23 : CC2=.54
2060 FOR I=1 TO NH
2070 X(N2+I)=(X(0)+X(1))*I
2080 X(N+1)=(X(N2-1)+X(N2))*I
2090 FOR J=2 TO N2
2100 X(N2+J)=CC1*(X(J-2)+X(J))+CC2*X(J-1)
2110 NEXT J
2120 FOR J=0 TO N2 : X(J)=X(J+N2+1) : NEXT
2130 NEXT I
2140 RETURN
Ok

```

## Program 7

```

LIST
1000 '*** SUB BURG ***
1010 SUM=0 : FOR I=1 TO N : SUM=SUM+X(I)^2 : NEXT
1020 C(1)=SUM/N : PM=C(1) : FPE(1)=(N+1)/(N-1)*PM
1030 LG1=LAG+1 : LGM1=LAG-1 : B1(1)=X(1)
1040 FOR I=2 TO N : B1(I)=X(I) : B2(I-1)=X(I) : NEXT
1050 FOR M=1 TO LGM1
1060 GOSUB 2000 : SUM=0
1070 FOR I=1 TO M : SUM=SUM+C(M-I+1)*A(I) : NEXT
1080 C(M+1)=SUM
1090 FPE(M+1)=(N+M+1)/(N-M-1)*PM : NEXT M
1100 G(1)=1 : FOR J=2 TO LAG : G(J)=-A(J-1) : NEXT
1110 IF LAG=LMAX THEN 3000
1120 FOR L=LG1 TO LMAX : SUM=0
1130 FOR I=2 TO LAG : SUM=SUM-C(L-I+1)*G(I) : NEXT
1140 C(L)=SUM : NEXT L
1150 GOTO 3000
2000 '*** SUB LEVINS ***
2010 STN=0 : STD=0 : IM=N-M
2020 FOR I=1 TO IM : STN=STN+B1(I)*B2(I) : STD=STD+B1(I)^2+B2(I)^2 : NEXT
2030 A(M)=2*STN/STD : PM=PM*(1-A(M)^2)
2040 IF M=1 THEN 2060
2050 MM1=M-1 : FOR K=1 TO MM1 : A(K)=G(K)-A(M)*G(M-K) : NEXT
2060 FOR I=1 TO M : G(I)=A(I) : NEXT : IMM=N-M-1
2070 FOR I=1 TO IMM : B1(I)=B1(I)-G(M)*B2(I) : B2(I)=B2(I+1)-G(M)*B1(I+1) : NEXT
2075 BEEP : CLS : PRINT "Go on G(I) ! I=";M
2080 RETURN
3000 '*** SUB MEM ***
3010 PMDT=PM*DT : FFN=FMIN*DT : FFX=FMAX*DT
3020 IF FFX>.5 THEN FFX=.5
3030 DFF=DF*DT : K=INT((FFX-FFN)/DFF+1.5)
3040 FOR I=1 TO K : FR=FFN+DFF*(I-1) : GOSUB 4000
3050 S(I)=PMDT/(CC^2+SS^2) : F(I)=FR/DT
3055 BEEP : CLS : PRINT "Go on S(I) ! I=";I
3060 NEXT I : RETURN
4000 '*** SUB FTRANS ***
4010 W#=FR*6.283185307179589#
4020 CW#=COS(W#) : SW#=SIN(W#) : CW2#=2*CW# : U#=0 : UV#=0 : Z=LAG
4030 FOR Q=1 TO LAG-1 : UVU#=UV# : UV#=U# : U#=G(Z)+UVU#*CW2#-UVU# : Z=Z-1 : NEXT
4040 CC=G(1)+U#*CW#-UV# : SS=U#*SW#
4050 RETURN
Ok

```

Program 8

```

LIST
10 *** TEST OF MEM ***
20 DIM X(25),B1(25),B2(25),G(25),C(25),FPE(25)
30 DIM A(11),F(51),S(51)
40 READ N,LAG,LMAX,FMIN,FMAX,DF,DT
50 DATA 25,11,15 ,0 ,.5 ,.01,1
60 P1=.942478 : P2=1.25664
70 FOR I=1 TO N
80 R=1.56*RND-.78
90 X(I)=R+2.6*COS(P1*I)+5.2*COS(P2*I)
100 NEXT I
110 SX=0 : FOR I=1 TO N : SX=SX+X(I) : NEXT
120 XM=SX/N : FOR I=1 TO N : X(I)=X(I)-XM : NEXT
130 CHAIN MERGE "MEM.SUB",140,ALL
140 GOSUB 1000
150 OPEN "2:MEM.RES" FOR OUTPUT AS #1
160 FOR I=1 TO K : PRINT #1,S(I) : NEXT
170 LPRINT :LPRINT "I","FREQ","SPEC","FPE","A.C."
180 FOR I=1 TO LAG : LPRINT I,F(I),S(I),FPE(I),C(I) : NEXT
190 FOR I=LAG+1 TO LMAX : LPRINT I,F(I),S(I) ,C(I) : NEXT
200 FOR I=LMAX+1 TO 30 : LPRINT I,F(I),S(I) : NEXT
210 LPRINT :LPRINT "----- The rest is omitted. -----"
220 END
Ok
RUN
    
```

I	FREQ	SPEC	FPE	A.C.
1	0	.841208	19.5693	18.064
2	.01	.684931	18.9459	5.89674
3	.02	.453186	2.53242	-13.1866
4	.03	.30823	2.6834	-14.1609
5	.04	.231555	2.78072	3.66384
6	.05	.193485	2.80666	14.8402
7	.06	.178891	1.70433	4.87582
8	.07	.182324	1.73453	-10.8422
9	.08	.205018	1.59984	-9.68787
10	.09	.256181	1.62862	5.92557
11	.1	.361562	1.79411	12.4813
12	.11	.595206		.0990372
13	.12	1.2214		-13.444
14	.13	3.68049		-7.82285
15	.14	29.2893		9.82749
16	.15	44.8082		
17	.16	8.93438		
18	.17	5.37631		
19	.18	5.64155		
20	.19	10.8872		
21	.2	133.82		
22	.21	21.391		
23	.22	2.98847		
24	.23	1.04905		
25	.24	.523901		
26	.25	.320475		
27	.26	.225452		
28	.27	.176466		
29	.28	.150621		
30	.29	.138239		

----- The rest is omitted. -----

## Program 9

```

LIST
1000 '*** FLTCOE.SUB ***
1010 'common PAS$,CF,HF,DT,D0,KK%,UU,A(),B()
1020 '*****
1030 P1=3.14159 : P2=P1/2
1040 DEF FN$H(F)=.5*(EXP(F)-EXP(-F))
1050 DEF FN$C(F)=.5*(EXP(F)+EXP(-F))
1060 IF PAS$="L" THEN 1070 ELSE IF PAS$="H" THEN 1080
1070 JP1=2 : JP2=1 : GOTO 1090
1080 JP1=1 : JP2=2
1090 F1=P1*CF*DT : F2=P1*HF*DT
1100 C=1/TAN(F1) : S=C*TAN(F2)
1110 ON JP1 GOTO 1120,1130
1120 S=1/S
1130 R=LOG(S+SQR(S*S-1))
1140 IF SGN(D0)<0 THEN 1150 ELSE 1160
1150 E2=10^(-.1*D0)-1 : GOTO 1170
1160 E2=D0
1170 E0=SQR(E2) : L%=INT(LOG((1+SQR(1-E2))/E0)/R)+1
1180 IF L%>12 THEN L%=12
1190 CH=FN$C(R*L%) : E=1/CH : UU=CH*2^(1-L%)
1200 P=LOG((1+SQR(1+E*E))/E)/L%
1210 SP=FN$H(P) : CP=FN$C(P) : KK%=L%*2
1220 IF L% MOD 2=0 THEN 1230 ELSE 1240
1230 JP3=1 : JP4=1 : GOTO 1250
1240 JP3=2 : JP4=2
1250 ON JP2 GOTO 1260,1350
1260 FOR I=1 TO KK%
1270 A(1,I)=1 : A(2,I)=2 : A(3,I)=1
1280 PP=P2*(2*I-1)/L% : Q=SIN(PP)*SP : O=COS(PP)*CP
1290 B(1,I)=(C+Q)^2+0^2 : B(2,I)=2*(Q^2+0^2-C^2) : B(3,I)=B(1,I)-4*C*Q
1300 NEXT I
1310 ON JP3 GOTO 1320,1330
1320 RETURN
1330 KK%=KK%+1 : A(1,KK%)=1 : A(2,KK%)=1 : A(3,KK%)=0
1340 B(1,KK%)=C+SP : B(2,KK%)=SP-C : B(3,KK%)=0 : RETURN
1350 CR=1/C
1360 FOR I=1 TO KK%
1370 A(1,I)=1 : A(2,I)=-2 : A(3,I)=1
1380 PP=P2*(2*I-1)/L% : Q=SIN(PP)*SP : O=COS(PP)*CP
1390 B(1,I)=(CR+Q)^2+0^2 : B(2,I)=2*(CR^2-Q^2-0^2) : B(3,I)=B(1,I)-4*CR*Q
1400 NEXT I
1410 ON JP4 GOTO 1420,1430
1420 RETURN
1430 KK%=KK%+1 : A(1,KK%)=1 : A(2,KK%)=-1 : A(3,KK%)=0
1440 B(1,KK%)=CR+SP : B(2,KK%)=CR-SP : B(3,KK%)=0
1450 RETURN
ok

```

## Program 10

```

LIST
3000 '*** FRERES.SUB ***
3010 'common A(),B(),RESP(),PHASE(),DT,KK%,UU,N
3020 PI=3.14159 : P12=PI*2 : P13=DT*PI2 : P2=PI/2 : X0=1E-30
3030 FM=.5/DT : DF=FM/(N-1)
3040 FOR I=1 TO N
3050 W=DF*(I-1)*PI3 : CS=COS(W) : SN=SIN(W) : R=1 : P=0
3060 FOR J=1 TO KK%
3070 X=A(2,J)+(A(1,J)+A(3,J))*CS : Y=(A(1,J)-A(3,J))*SN
3080 R=R*(X^2+Y^2) : GOSUB 3200 : P=P+AT
3090 X=B(2,J)+(B(1,J)+B(3,J))*CS : Y=(B(1,J)-B(3,J))*SN
3100 R=R/(X^2+Y^2) : GOSUB 3200 : P=P-AT
3110 ON SGN(P)+2 GOTO 3120,3160,3140
3120 ON SGN(P+PI)+2 GOTO 3130,3160,3160
3130 P=P+PI2 : GOTO 3110
3140 ON SGN(P-PI)+2 GOTO 3160,3160,3150
3150 P=P-PI2 : GOTO 3110
3160 NEXT J
3170 RESP(I)=R*UU : PHASE(I)=P/PI2
3180 NEXT I
3190 RETURN
3200 '*** SUB ATAN ***
3210 IF ABS(X)<X0 AND Y>0 THEN AT=P2 : RETURN
3220 IF ABS(X)<X0 AND Y<0 THEN AT=-P2 : RETURN
3230 IF X<0 AND Y>0 THEN AT=PI+ATN(Y/X) : RETURN
3240 IF X<0 AND Y<0 THEN AT=-PI+ATN(Y/X) : RETURN
3250 AT=ATN(Y/X) : RETURN
Ok

```

## Program 11

```

LIST
2000 '*** CASCAD.SUB ***
2010 'common X(),N,A(),B(),KK%,UU,DIR$
2020 DIM W(12)
2030 IF DIR$="P" THEN J=1:K=1 ELSE J=N:K=-1
2040 L=KK%*2
2050 FOR I=1 TO L : W(I)=0 : NEXT
2060 FOR I=1 TO KK%
2070 A(1,I)=A(1,I)/B(1,I):A(2,I)=A(2,I)/B(1,I):A(3,I)=A(3,I)/B(1,I):B(2,I)=B(2,I)/B(1,I):B(3,I)=B(3,I)/B(1,I)
2080 NEXT I
2090 X1=0 : X2=0
2100 FOR II=1 TO N
2110 X3=X2:X2=X1:X1=X(J):U3=X3:U2=X2:U1=X1:L=L+1
2120 FOR I=1 TO KK%
2130 U3=W(L+1) : U2=W(L)
2140 U1=A(1,I)*U1+A(2,I)*U2+A(3,I)*U3-B(2,I)*U2-B(3,I)*U3
2150 W(L)=U1:W(L+1)=U2:U1=U1:U2=U2:U3=U3:L=L+2
2160 NEXT I
2170 X(J)=U1*UU : J=J+K : NEXT II
2180 RETURN
Ok


```

## Program 12

```

LIST
10 '*** Test of filtering Program ***
20 COMMON PAS$,CF, HF, DT, D0, KK%, UV, A(), B(), RESP(), PHASE(), N
30 DIM A(3,6), B(3,6), RESP(251), PHASE(251)
40 READ PAS$, N, CF, HF, DT, D0, KK%, UV
50 DATA L, 251, 60, 62, .002, -.5, 0, 0
60 CHAIN MERGE "FLTCOE.SUB", 70
70 GOSUB 1000 : '--- FLTCOE.SUB
80 CHAIN MERGE "FRERES.SUB", 90
90 GOSUB 3000 : '--- FRERES.SUB
95 LPRINT "KK="; KK%, "UV="; UV, "DF="; DF
100 OPEN "2:RESP.DAT" FOR OUTPUT AS #1
110 FOR I=1 TO N :PRINT #1, RESP(I); :NEXT
120 OPEN "2:PHASE.DAT" FOR OUTPUT AS #2
130 FOR I=1 TO N :PRINT #2, PHASE(I); :NEXT
140 END
Ok

```



第 1 図

Program 13

```

LIST 10-500
10 '*** HINDO ***
20 DEFINT I,X,Y,Z,K,N
30 INPUT "File name of data ";FIL$
40 INPUT "Number of data ";N
50 INPUT "Name of data ";F$
60 INPUT "Name of unit ";U$
70 N=N-1 : DIM A(N)
80 OPEN FIL$ FOR INPUT AS #1
90 FOR I=0 TO N : INPUT #1,A(I) : NEXT
100 '*** Calculation ***
110 IF A(0)>A(1) THEN AMAX=A(0):AMIN=A(1) ELSE AMAX=A(1):AMIN=A(0)
120 FOR I=2 TO N
130 IF A(I)>AMAX THEN AMAX=A(I) ELSE IF A(I)<AMIN THEN AMIN=A(I)
140 NEXT I
150 AA=AMAX : GOSUB 600 : IF F2=0 THEN 180
160 FOR I=0 TO N : A(I)=A(I)*F1 : NEXT I
170 AMAX=AMAX*F1 : AMIN=AMIN*F1
180 W=20*(AMAX-AMIN)/N+.05 : W=INT(W*10)/10 : W2=W/2
190 Y=INT(AMIN/W)+1 : Z=INT(AMAX/W)+1 : IE=Z-Y
200 DIM K(IE),P(IE)
210 M21=0 : M22=0 : M23=0
220 FOR I=0 TO IE : K(I)=0 : NEXT
230 FOR I=0 TO N : X=INT(A(I)/W)+1-Y : K(X)=K(X)+1 : NEXT : IMAX=1
240 KMAX=K(0) : FOR I=1 TO IE : IF KMAX<K(I) THEN KMAX=K(I):IMAX=I:NEXT
250 KSUM=0 : FOR I=0 TO IE : KSUM=KSUM+K(I) : NEXT
260 FOR I=0 TO IE : P(I)=K(I)/KSUM*100 : NEXT
270 FOR I=0 TO IE : II=I-IMAX
280 M21=M21+II*K(I) : M22=M22+II^2*K(I) : M23=M23+II^3*K(I)
290 NEXT
300 M21=M21*W/KSUM : M=M21+(IMAX+Y)*W-W2 : M22=M22*W^2/KSUM
310 M23=M23*W^3/KSUM : M2=M22-M21^2 : RMS=SQR(M2)
320 M3=M23-3*M22*M21+2*M21^3 : SK=M3/RMS^3
330 '*** Print of result ***
340 DEF CHR$(&HA5)="000000C0C0000000"
350 D$=CHR$(&HA5) : C$=CHR$(&HF1) : S$=STRING$(9," ")
360 LPRINT "Frequency distribution of ";F$
370 LPRINT:LPRINT "Unit=E";:LPRINT USING "+# ";F2;:LPRINT U$ : LPRINT SPC(11)
380 LPRINT "0      10      20      30      40      50%"
390 LPRINT USING "MIN:###.## ";AMIN;
400 LPRINT " |          |          |          |          | "
410 FOR I=0 TO IE
420 U=W*(I+Y)-W2 : IP=INT(P(I)+.5) : U$=STRING$(IP,CHR$(&HA0))
430 LPRINT USING "    ###.## ";U;
440 IF IP< 1 THEN LPRINT C$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$ SPC(4);:GOTO 520
450 IF IP<= 9 THEN LPRINT U$SPC(10-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
460 IF IP<=19 THEN LPRINT U$SPC(20-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
470 IF IP<=29 THEN LPRINT U$SPC(30-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
480 IF IP<=39 THEN LPRINT U$SPC(40-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
490 IF IP<=49 THEN LPRINT U$SPC(50-IP)D$+S$+D$+S$+D$+S$+D$+S$+D$+S$+D$SPC(4);:GOTO 520
500 IF IP =50 THEN LPRINT U$SPC(5);:GOTO 520

```

Ok

## Program 13の続き

```

LIST 510-
510 IF IP> 50 THEN LPRINT STRING$(50,CHR$(&HR0))+"Over ";
520 LPRINT USING "###.#";P(I); : LPRINT USING "    ###";K(I)
530 NEXT I
540 LPRINT USING "MAX:###.## ";AMAX;
550 LPRINT " | | | | | "
560 LPRINT TAB(69)"Total=";KSUM
570 LPRINT : LPRINT TAB(11)"Mean=";M;" RMS=";RMS;" Skewness=";SK
580 END
600 '*** SUB F1,F2 ***
610 F2=0
620 IF AA>=100 THEN 650 ELSE IF AA>=10 THEN 670 ELSE 630
630 AA=AA*10 : F2=F2-1
640 IF AA<10 THEN 630 ELSE 670
650 AA=AA/10 : F2=F2+1
660 IF AA>=100 THEN 650 ELSE 670
670 F1=10^-F2
680 RETURN
Ok

```

## Frequency distribution of Y12.DAT(1-250)

Unit=E-1 m/s

	0	10	20	30	40	50%		
MIN:	2.44							
	2.15 █████	.	.	.	.	.	5.2	13
	6.45 ██████████	.	.	.	.	.	9.6	24
	10.75 ██████████████	.	.	.	.	.	12.8	32
	15.05 ██████████████████	.	.	.	.	.	17.6	44
	19.35 ████████████████████	.	.	.	.	.	16.0	40
	23.65 ██████████████████████	.	.	.	.	.	12.0	30
	27.95 ████████████████████████	.	.	.	.	.	10.0	25
	32.25 ██████████████████████████	.	.	.	.	.	6.4	16
	36.55 ████████████████████████████	.	.	.	.	.	3.6	9
	40.85 ██████████████████████████████	.	.	.	.	.	4.0	10
	45.15 ████████████████████████████████	.	.	.	.	.	1.6	4
	49.45	.	.	.	.	.	0.4	1
	53.75	.	.	.	.	.	0.4	1
	58.05	.	.	.	.	.	0.4	1
MAX:	56.28							

Total= 250

Mean= 19.866 RMS= 10.8728 Skewness= .638389

## Program 14

```
LIST
10 '*** DATA.GRA ***
20 DIM X(S12) : CLS
30 INPUT "File name of data";A$
40 INPUT "Numbers of data and interval";N,DT
45 INPUT "Unit of interval";U$
50 OPEN A$ FOR INPUT AS #1
60 FOR I=0 TO N-1 : INPUT #1,X(I) : NEXT
70 IF X(0)<=X(1) THEN XN=X(0) : XM=X(1) ELSE XN=X(1) : XM=X(0)
80 FOR I=2 TO N-1
81 IF X(I)<XN THEN XN=X(I) ELSE IF X(I)>XM THEN XM=X(I)
82 NEXT I
90 XMN=XM-XN : DEF FNY(F)=20+160*(XM-F)/XMN
100 CLS : INPUT "Numbers of start and end";S,E
110 CLS
120 LINE(0,20)-(0,180),PSET
130 IF XM<0 OR XN>0 THEN Y0=180 ELSE Y0=FNY(0)
140 LINE(0,Y0)-(639,Y0),PSET
150 X1=0 : Y1=FNY(X(S)) : ST=639/(E-S)
160 FOR I=S+1 TO E : X2=X1+ST : Y2=FNY(X(I))
170 LINE(X1,Y1)-(X2,Y2),PSET : X1=X2 : Y1=Y2 : NEXT
180 Z=XM:GOSUB 300 : Z=XN:GOSUB 300 : IF XM<0 OR XN>0 THEN 195
190 Z=0 : GOSUB 300
195 MY=INT(Y0/8)+1
200 FOR I=1 TO 4 :PX=127.8*I:MX=INT(PX/8+2):Z=(S+(E-S)/5*I)*DT
210 LINE(PX,Y0)-(PX,Y0+3),PSET:LOCATE MX,MY:PRINT MID$(STR$(Z),2,4)
215 NEXT I
220 LOCATE 76,MY : PRINT U$
230 LOCATE 0,0 : INPUT "Continue or Copy (Y/N/C)";B$
235 IF B$="C" THEN CLS 0,1:LOCATE 0,0:PRINT A$:COPY:CLS 0,1:GOTO 230
240 IF B$="Y" THEN GOTO 100 ELSE END
300 PY=FNY(Z) : MY=INT(PY/8)
310 LINE(0,PY)-(3,PY),PSET
320 LOCATE 1,MY:IF Z=0 THEN PRINT Z ELSE IF ABS(Z)<1 THEN PRINT USING "#.###";Z
ELSE PRINT USING "###.##";Z
330 RETURN
Ok
```


## Program 15

```

LIST
10 '*** DATA2.GRA ***
20 DIM X(512),Y(512) : CLS
30 INPUT "File name of 2-data";A$,B$
40 INPUT "Number of data and interval";N,DT
50 INPUT "Unit of interval";U$
60 OPEN A$ FOR INPUT AS #1
70 OPEN B$ FOR INPUT AS #2
80 FOR I=0 TO N-1 : INPUT #1,X(I) : NEXT
90 FOR I=0 TO N-1 : INPUT #2,Y(I) : NEXT
100 IF X(0)<=X(1) THEN XN=X(0) : XM=X(1) ELSE XN=X(1) : XM=X(0)
110 FOR I=2 TO N-1
120 IF X(I)<XN THEN XN=X(I) ELSE IF X(I)>XM THEN XM=X(I)
130 NEXT I
140 IF Y(0)<=Y(1) THEN YN=Y(0) : YM=Y(1) ELSE YN=Y(1) : YM=Y(0)
150 FOR I=2 TO N-1
160 IF Y(I)<YN THEN YN=Y(I) ELSE IF Y(I)>YM THEN YM=Y(I)
170 NEXT I
180 XMN=XM-XN : DEF FNX(F)= 20+72*(XM-F)/XMN
190 YMN=YM-YN : DEF FNY(F)=100+72*(YM-F)/YMN
195 DEF FNM(F)=INT((F-4)/8)+1
200 CLS : INPUT "Number of start and end";S,E : CLS
210 LINE(0,20)-(0,92),PSET
220 IF XM<0 OR XN>0 THEN X0=92 ELSE X0=FNX(0)
230 LINE(0,X0)-(639,X0),PSET
240 LINE(0,100)-(0,180),PSET
250 IF YM<0 OR YN>0 THEN Y0=180 ELSE Y0=FNY(0)
260 LINE(0,Y0)-(639,Y0),PSET
270 ST=639/(E-5) : X1=0 : Y1=FNX(X(S))
280 FOR I=S+1 TO E : X2=X1+ST : Y2=FNX(X(I))
290 LINE(X1,Y1)-(X2,Y2),PSET : X1=X2 : Y1=Y2 : NEXT I
300 X1=0 : Y1=FNY(Y(S))
310 FOR I=S+1 TO E : X2=X1+ST : Y2=FNY(Y(I))
320 LINE(X1,Y1)-(X2,Y2),PSET : X1=X2 : Y1=Y2 : NEXT I
330 Z=XM:PY=20:MY=2 : GOSUB 1000 : Z=XN:PY=92:MY=11 : GOSUB 1000
340 IF XM>0 AND XN<0 THEN MY=FNM(X0):LOCATE 1,MY:PRINT "0"
350 PY=X0 : IF PY=92 THEN MY=12 ELSE MY=FNM(X0)+1
355 GOSUB 2000
360 Z=YM:PY=100:MY=13 : GOSUB 1000 : Z=YN:PY=180:MY=22 : GOSUB 1000
370 IF YM>0 AND YN<0 THEN MY=FNM(Y0):LOCATE 1,MY:PRINT "0"
380 PY=Y0 : IF PY=180 THEN MY=23 ELSE MY=FNM(Y0)+1
385 GOSUB 2000
390 LOCATE 0,0 : INPUT "Continue or Copy (Y/N/C)";C$
400 IF C$="C" THEN CLS 0,1:LOCATE 0,1:PRINT A$:LOCATE 0,12:PRINT B$:COPY:GOTO 39
0
410 IF C$="Y" THEN 200 ELSE CLS : END
1000 LINE(0,PY)-(3,PY),PSET
1010 LOCATE 1,MY:IF ABS(Z)<1 THEN PRINT USING "#.###";Z ELSE PRINT USING "###.##
";Z
1020 RETURN
2000 FOR I=1 TO 4 : PX=127.8*I : MX=INT(PX/8+1) : Z=(S+(E-S)/5*I)*DT
2010 LINE(PX,PY)-(PX,PY+3),PSET : LOCATE MX,MY:PRINT MID$(STR$(Z),2,4)
2020 NEXT I : LOCATE 76,MY : PRINT U$
2030 RETURN
Ok


```

2:NEW.RES



第 3 図

2:NEW.RES



第 4 図

## Program 16

```

LIST
10 '*** Graph of Power Spectrum *** "SPECTR.GRA"
20 DIM P(512) : CLS
30 INPUT "File name of data":A$
40 INPUT "Numbers of data and interval":N,DF
50 OPEN A$ FOR INPUT AS #1
60 FOR I=0 TO N-1 : INPUT #1,P(I) : NEXT
70 PM=P(0)
75 FOR I=0 TO N-1
76 IF P(I)>PM THEN PM=P(I)
77 IF P(I)<1E-10 THEN P(I)=1E-10
78 NEXT I
80 CLS : INPUT "Numbers of start and end":S,E
90 IF S=0 THEN S=1
95 FS=DF*S : FE=DF*E
100 B=639/LOG(E/S) : C=160/LOG(1000) : C1=24/197 : C2=80/639
110 X=PM : GOSUB 3000
120 IF X>5 THEN ZM=1*10^(P+1) : ZN=ZM/1000 : K=1 : GOTO 150
130 IF X>2 THEN ZM=5*10^P : ZN=ZM/1000 : K=2 : GOTO 150
140 ZM=2*10^P : ZN=ZM/1000 : K=3
150 DEF FN X(F)=B*LOG(F/FS)
160 DEF FN Y(F)=20+C*LOG(ZM/F)
170 CLS : LINE(0,20)-(0,180),PSET
180 ON K GOTO 190,200,210
190 Z=1*10^(P+1) : IF Z<ZN THEN 230 ELSE GOSUB 1000
200 Z=5*10^P : IF Z<ZN THEN 230 ELSE GOSUB 1000
210 Z=2*10^P : IF Z<ZN THEN 230 ELSE GOSUB 1000
220 P=P-1 : GOTO 190
230 LINE(0,180)-(639,180),PSET
240 X=FS : GOSUB 3000
250 IF X<2 THEN 260 ELSE IF X<5 THEN 270 ELSE 280
260 Z=1*10^P : IF Z>FE THEN 300 ELSE GOSUB 2000
270 Z=2*10^P : IF Z>FE THEN 300 ELSE GOSUB 2000
280 Z=5*10^P : IF Z>FE THEN 300 ELSE GOSUB 2000
290 P=P+1 : GOTO 260
300 X0=0 : Y0=FN Y(P(S)) : DRAW "BM=X0;,=Y0;"
310 FOR I=S+1 TO E : X1=FN X(DF*I)
320 IF P(I)<=ZN THEN Y1=180 ELSE Y1=FN Y(P(I))
330 DRAW "M=X1;,=Y1;"
340 NEXT I
350 LOCATE 70,21 : PRINT "Hz"
355 LOCATE 0,0 : INPUT "Continue or Copy (Y/N/C)":B$
356 IF B$="C" THEN CLS 0,1 : LOCATE 0,0 : PRINT A$ : COPY : CLS 0,1 : GOTO 355
360 IF B$="Y" THEN GOTO 80 ELSE END
1000 '*** Y-AXIS ***
1010 PY=FN Y(Z) : MY=INT(C1*PY)
1020 LINE(0,PY)-(3,PY),PSET
1030 IF MY<=22 THEN LOCATE 1,MY : PRINT Z
1040 RETURN
2000 '*** X-AXIS ***
2010 PX=FN X(Z) : IF PX<0 THEN RETURN ELSE MX=INT(C2*PX)
2020 LINE(PX,180)-(PX,183),PSET
2030 IF MX<75 THEN LOCATE MX,23 : PRINT Z
2040 RETURN
3000 '*** POWER SEARCH ***
3010 P=0
3020 IF X>=10 THEN 3030 ELSE IF X>=1 THEN RETURN ELSE 3050
3030 X=X/10 : P=P+1
3040 IF X>=10 THEN 3030 ELSE RETURN
3050 X=X*10 : P=P-1
3060 IF X<1 THEN 3050 ELSE RETURN
Ok

```

## Program 17

```

LIST 10-400
10 '*** Contour ***
20 DEFINT I,J,K,L,M,N,B
30 INPUT "Size of data (Tate,Yoko:Yoko<Tate*1.33) ";N,M
40 INPUT "File name of data ";F$
50 INPUT "Maximum and minimum value of data ";AMAX,AMIN
60 INPUT "Number of contour-line between MAX and MIN (<=30) ";ND
65 MN=ND+1 : NB=ND+2
70 OPTION BASE 1 : DIM A(M,N),B(4,NB),X(4,NB),Y(4,NB)
80 OPEN F$ FOR INPUT AS #1
90 FOR I=1 TO M:FOR J=1 TO N:INPUT #1,A(I,J):NEXT J,I
100 DA=(AMAX-AMIN)/ND
110 FOR I=1 TO M:FOR J=1 TO N:A(I,J)=A(I,J)-AMIN:NEXT J,I
120 D2=199/(N-1):D1=D2*2.41:D3=.5*D1:D4=.5*D2
130 CLS : BP=-1 : BQ=-1
140 FOR I=1 TO M-1
150 II=I-1:O1=II*D1:O2=O1+D1:O5=O1+D3
160 FOR J=1 TO N-1
170 O=.25*(A(I,J)+A(I+1,J)+A(I,J+1)+A(I+1,J+1))
180 JJ=J-1:O3=199-JJ*D2:O4=O3-D2:O6=O3-D4
190 K=1:A1=A(I,J):X1=O1:Y1=O3:A2=A(I,J+1):X2=O1:Y2=O4
200 GOSUB 1000
210 K=2:A1=A(I,J+1):X1=O1:Y1=O4:A2=O: X2=O5:Y2=O6
220 GOSUB 1000
230 K=3:A1=O:X1=O5:Y1=O6:A2=A(I,J):X2=O1:Y2=O3
240 GOSUB 1000
250 L1=1:L2=2:L3=3:GOSUB 2000
260 K=1:A1=A(I,J):X1=O1:Y1=O3:A2=A(I+1,J):X2=O2:Y2=O3
270 GOSUB 1000
280 K=4:A1=A(I+1,J):X1=O2:Y1=O3:A2=O: X2=O5:Y2=O6
290 GOSUB 1000
300 L1=3:L2=1:L3=4:GOSUB 2000
310 K=3:A1=O:X1=O5:Y1=O6:A2=A(I+1,J+1):X2=O2:Y2=O4
320 GOSUB 1000
330 K=1:A1=A(I+1,J+1):X1=O2:Y1=O4:A2=A(I+1,J):X2=O2:Y2=O3
340 GOSUB 1000
350 L1=4:L2=1:L3=3:GOSUB 2000
360 K=1:A1=A(I+1,J+1):X1=O2:Y1=O4:A2=A(I,J+1):X2=O1:Y2=O4
370 GOSUB 1000
380 L1=3:L2=2:L3=1:GOSUB 2000
390 NEXT J,I
395 LPRINT "*** ";F$;" ***"
396 LPRINT "MAX=";AMAX;" MIN=";AMIN;" N=";ND;" DA=";DA
397 COPY
400 END
Ok

```

## Program 17の続き


```

LIST 1000-
1000 '*** SUB1 ***
1010 A3=A2-A1:N1=INT(A1/DA+.5)
1020 FOR I0=1 TO NN:B(K,I0)=0:NEXT
1030 IF A3<0 THEN 1170 ELSE IF A3=0 THEN 1040 ELSE 1080
1040 DAN=DA*N1:IF A1<DAN THEN RETURN
1050 I1=2
1060 N2=N1+1:B(K,N2)=-1:X(K,N2)=X1:Y(K,N2)=Y1
1070 ON I1 GOTO 1090,1240,1180
1080 DAN=DA*N1:IF A1<DAN THEN 1100 ELSE IF A1=DAN THEN 1160 ELSE 1090
1090 N1=N1+1
1100 N2=N1+1
1110 DAN=DA*N1:IF A2<DAN THEN RETURN ELSE IF A2=DAN THEN 1230 ELSE 1120
1120 I2=1
1130 B(K,N2)=-1:F=(DA*N1-A1)/A3
1140 X(K,N2)=X1+(X2-X1)*F:Y(K,N2)=Y1+(Y2-Y1)*F
1150 ON I2 GOTO 1090,1180
1160 I1=1:GOTO 1060
1170 DAN=DA*N1:IF A1<DAN THEN 1180 ELSE IF A1=DAN THEN 1210 ELSE 1190
1180 N1=N1-1
1190 N2=N1+1
1200 DAN=DA*N1:IF A2<DAN THEN 1220 ELSE IF A2=DAN THEN 1230 ELSE RETURN
1210 I1=3:GOTO 1060
1220 I2=2:GOTO 1130
1230 B(K,N2)=-1:X(K,N2)=X2:Y(K,N2)=Y2
1240 RETURN
2000 '*** SUB2 ***
2010 FOR I0=1 TO NN
2015 IF I0=1 AND BP OR I0=NN AND B0 THEN BB=-1 ELSE BB=0
2020 IF NOT B(L1,I0) AND NOT B(L2,I0) OR NOT B(L1,I0) AND NOT B(L3,I0) OR NOT B(L2,I0) AND NOT B(L3,I0) THEN 2170
2030 IF B(L1,I0) AND B(L2,I0) AND B(L3,I0) THEN 2070
2040 IF B(L1,I0) AND B(L2,I0) THEN 2110
2050 IF B(L1,I0) AND B(L3,I0) THEN 2130
2060 IF B(L2,I0) AND B(L3,I0) THEN 2150
2070 DRAW"BM=X(L1,I0);,=Y(L1,I0);"
2080 DRAW "M=X(L2,I0);,=Y(L2,I0);"
2095 IF BB THEN 3000
2090 DRAW "M=X(L3,I0);,=Y(L3,I0);"
2100 DRAW "M=X(L1,I0);,=Y(L1,I0);":GOTO 2170
2110 DRAW"BM=X(L1,I0);,=Y(L1,I0);"
2120 DRAW "M=X(L2,I0);,=Y(L2,I0);":GOTO 2170
2125 IF BB THEN 3000
2130 DRAW"BM=X(L1,I0);,=Y(L1,I0);"
2140 DRAW "M=X(L3,I0);,=Y(L3,I0);":GOTO 2170
2150 DRAW"BM=X(L2,I0);,=Y(L2,I0);"
2155 IF BB THEN 3000
2160 DRAW "M=X(L3,I0);,=Y(L3,I0);"
2170 NEXT I0
2180 RETURN
3000 IX=INT(.123631*X(L2,I0)):IY=INT(.120603*Y(L2,I0))
3010 LOCATE IX,IY
3015 IF I0=1 THEN PRINT AMIN:BP=0 ELSE PRINT AMAX:B0=0
3020 BB=0 : RETURN
ok

```

J/I	1	2	3	4	5	6	7	8
8	3.36	3.04	3.51	3.78	3.03	3.26	3.50	4.20
7	4.50	4.96	4.87	4.29	4.36	2.75	3.28	4.02
6	4.79	5.00	5.86	5.78	4.57	2.23	3.04	4.88
5	4.06	5.97	6.91	5.60	4.86	2.37	3.00	3.38
4	4.52	5.01	5.82	5.03	4.88	2.61	2.89	2.13
3	4.31	4.73	4.55	4.87	4.22	0.20	0.04	0.73
2	3.31	3.90	3.66	3.45	2.23	0.76	-0.55	-0.62
1	2.25	2.98	2.90	2.58	2.14	0.88	-0.07	-1.25

\*\*\* 2:CONTR4.DAT \*\*\*  
 MAX= 6 MIN=-1 N= 10 DR= .7



第 5 図

## Program 18

```

LIST
10 '*** Test of FFT ***
20 DEFINT I-N
30 DIM X(16),A(8),B(8),S(4)
40 COMMON X(),A(),B(),S()
50 COMMON N,N2,N4,NP
60 READ N,N2,N4,NP
70 DATA 16, 8, 4, 4
80 FOR I=1 TO 16 :READ X(I) :NEXT
90 DATA 0,1,2,3,4,5,6,7,8,7,6,5,4,3,2,1
100 FOR I=1 TO N4-1 :S(I)=SIN(I*6.28319/N) :NEXT I
110 CHAIN MERGE "FFT.SUB",120
120 T=TIME :GOSUB 1000 : '--- FFT.SUB
130 PRINT "Time=";TIME-T;"sec" :PRINT
140 PRINT "I","A","B"
150 FOR I=1 TO 8 :PRINT I,A(I),B(I) :NEXT I
160 END
Ok
RUN
Time= 3 sec

```

I	A	B
1	64	0
2	-26.2742	-1.43051E-06
3	0	0
4	-3.23983	4.23193E-06
5	0	0
6	-1.44646	4.23193E-06
7	0	0
8	-1.03956	-1.43051E-06

Ok

Program 19

```

LIST
10 '*** Test of FTRAN ***
20 DEFINT I,J,L,Z,Q
30 DIM G(16),A(9),B(9)
50 READ LAG,OFF
60 DATA 16,.0625
70 FOR I=1 TO 16 :READ G(I) :NEXT
80 DATA 0,1,2,3,4,5,6,7,8,7,6,5,4,3,2,1
90 CHAIN MERGE "FTRAN.SUB",100,ALL
100 T=TIME
110 FOR I=1 TO 9 :FR=OFF*(I-1)
120 GOSUB 4000 : '--- FTRAN.SUB
130 A(I)=CC :B(I)=SS :NEXT I
140 PRINT "Time=":TIME-T;"sec" :PRINT
150 PRINT "I","A","B"
160 FOR I=1 TO 9 :PRINT I,A(I),B(I) :NEXT
170 END
Ok
RUN
Time= 9 sec
    
```

I	A	B
1	64	0
2	-26.2741	-5.50624E-14
3	9.52421E-18	-1.57009E-16
4	-3.23983	-2.01553E-14
5	5.32443E-32	-5.55112E-17
6	-1.44646	-1.41548E-14
7	2.41671E-16	-1.96262E-17
8	-1.03957	-1.4934E-14
9	0	0

Ok